# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=lsp-test
pkgname=haskell-lsp-test0.14.0
pkgver=0.14.0.2
pkgrel=71
pkgdesc="Functional test framework for LSP servers."
url="https://github.com/haskell/lsp/blob/master/lsp-test/README.md"
license=("BSD")
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-diff' 'haskell-glob' 'haskell-aeson' 'haskell-aeson-pretty'
         'haskell-ansi-terminal' 'haskell-async' 'haskell-conduit' 'haskell-conduit-parse'
         'haskell-data-default' 'haskell-lens' 'haskell-lsp1.4' 'haskell-lsp-types1.4'
         'haskell-parser-combinators' 'haskell-some' 'haskell-unliftio'
         'haskell-unordered-containers')
makedepends=('ghc' 'uusi' 'haskell-hspec')
source=("https://hackage.haskell.org/packages/archive/$_hkgname/$pkgver/$_hkgname-$pkgver.tar.gz")
sha512sums=('7b850649d3413a02cddf4be7d353a2f268d28662f0a996532a407804179b12714dd5fd521de8168be3a8556dc1543177b5b06ff6aa6379d482fb808227c30e10')

prepare() {
  cd $_hkgname-$pkgver
  gen-setup
  sed -i 's|lsp-demo-reactor-server|lsp-demo-reactor-server-1.4|g' example/Test.hs
}

build() {
  cd $_hkgname-$pkgver

  runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
    --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
    --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm \
    --ghc-option=-optl-Wl\,-z\,relro\,-z\,now \
    --ghc-option='-pie'

  runhaskell Setup build $MAKEFLAGS
  runhaskell Setup register --gen-script
  runhaskell Setup unregister --gen-script
  sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
  sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
  cd $_hkgname-$pkgver
  runhaskell Setup test --show-details=direct
}

package() {
  cd $_hkgname-$pkgver

  install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
  install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
  runhaskell Setup copy --destdir="$pkgdir"
  install -D -m644 LICENSE -t "$pkgdir"/usr/share/licenses/$pkgname/
  rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
